
                                                    29.09.2013 18:55:06
                                            

                                
                                            Paddey
                                    

            

        

        
                            
Hallo an alle:)

ich hab da ein kleineres Problem.

Erschlagt mich bitte nicht gleich wenn es jetzt nicht das schönste Makro ist was ihr bis jetzt gesehen habt, ist das erste Mal das ich mich mit dem Thema Makros auseinander gesetzt habe.

Das was ich da bis jetzt zusammen geschrieben habe, ist alles mit Suchen im Internet entstanden, doch nun komme ich nicht weiter.

Und zwar habe ich eine Tabelle mit 8 Mappen, wobei in den erste 5 Tabellenblätter Infos gesammelt werden und auf einer versteckten Mappe (7. Checkliste) zusammengetragen werden.

Nachdem alles ausgefüllt wurde soll dann der Anwender in der 6 Mappe(6. Speichern)

einen Button drücken und danach soll die versteckte Mappe (7. Checkliste) einmal per Pdf und einmal als .xls Datei (aber nur mit Werten, ohne Formeln) an eine E-Mail Adresse geschickt werden.

Und die Dateien sollen beide denselben Namen haben (siehe Formel).

Ich hoffe ihr könnt mir mit meinem Problem weiterhelfen.

Code:	
Sub Schaltfläche3_Klicken()
' Schaltfläche3_Klicken Makro
Sheets("7. Checkliste").Visible = True
Sheets("7. Checkliste").Select
ActiveSheet.AutoFilter.ApplyFilter
Sheets("6. Speichern").Select
Dim app   As Object
Dim file  As String
Dim isNew As Boolean
Sheets("7. Checkliste").Select
file = Left(ThisWorkbook.Name, Len(ThisWorkbook.Name) - 5) & " " & Worksheets("2. Auswahl"). _
Range("C4").Value & " " & Worksheets("2. Auswahl").Range("C6").Value & ".pdf"
ActiveSheet.ExportAsFixedFormat xlTypePDF, Environ("TEMP") & "\" & file
On Error Resume Next
Set app = GetObject(, "Outlook.Application")
If app Is Nothing Then
Set app = CreateObject("Outlook.Application")
isNew = True
End If
With app.CreateItem(0)
.To = "verbrauchsmaterial.it@test.de"
.CC = ""
.BCC = ""
.Subject = "Preinstallationsliste"
.Body = "Sehr geehrte Damen und Herren." & vbCr _
& vbCr _
& "Anbei die bearbeitete Preinstallationsliste." & vbCr _
& vbCr _
& "Mit freundlichen Grüßen."
.Attachments.Add Environ("TEMP") & "\" & file
.Send
End With
If isNew Then app.Quit
Dim WB As Workbook
For Each WB In Application.Workbooks
WB.Saved = True
Next
Application.Quit
End Sub




Es klappt eigentlich alle soweit bis auf der XLS Part.

Danke schonmal im Voraus

MfG

Paddey

Hallo Paddy,

hier dein Makro ergänzt um den Versand der Checkliste als xls-Datei.

Ich hab noch ein klein wenig aufgeräumt. 

Vergiß nicht die Datei zu speichern, bevor du den Mailversand testest. Dein Makro schließt ja alle Excel-Dateien inkl. Excel selbst ohne die Dateien zu speichern.

Gruß

Franz

Sub Schaltfläche3_Klicken()
' Schaltfläche3_Klicken Makro
Dim app   As Object
Dim filePDF  As String, fileXLS As String, file As String
Dim isNew As Boolean
Sheets("7. Checkliste").Visible = True
Sheets("7. Checkliste").Select
ActiveSheet.AutoFilter.ApplyFilter
Sheets("6. Speichern").Select
Sheets("7. Checkliste").Select
file = Left(ThisWorkbook.Name, Len(ThisWorkbook.Name) - 5) & " " _
& Worksheets("2. Auswahl").Range("C4").Value & " " _
& Worksheets("2. Auswahl").Range("C6").Value
fileXLS = Environ("TEMP") & "\" & file & ".xls"
filePDF = Environ("TEMP") & "\" & file & ".pdf"
ActiveSheet.ExportAsFixedFormat xlTypePDF, filePDF
'aktives Blatt ohne Formeln in separater Datei speichern
ActiveSheet.Copy
With ActiveWorkbook
With .Worksheets(1)
.UsedRange.Copy
.UsedRange.PasteSpecial Paste:=xlPasteValues
Application.CutCopyMode = False
End With
Application.DisplayAlerts = False
If Val(Left(Application.Version, 2)) >= 12 Then
'Excel 2007 und neuer
.SaveAs Filename:=fileXLS, FileFormat:=-4143 'xlWorkbookNormal
Else
'Excel 2003 und älter
.SaveAs Filename:=fileXLS
End If
.Close savechanges:=True
End With
Application.DisplayAlerts = False
On Error Resume Next
Set app = GetObject(, "Outlook.Application")
If app Is Nothing Then
Set app = CreateObject("Outlook.Application")
isNew = True
End If
With app.CreateItem(0)
.To = "verbrauchsmaterial.it@test.de"
.CC = ""
.BCC = ""
.Subject = "Preinstallationsliste"
.Body = "Sehr geehrte Damen und Herren." & vbCr _
& vbCr _
& "Anbei die bearbeitete Preinstallationsliste." & vbCr _
& vbCr _
& "Mit freundlichen Grüßen."
.Attachments.Add filePDF
.Attachments.Add fileXLS
.Send
End With
If isNew Then app.Quit
Dim WB As Workbook
For Each WB In Application.Workbooks
WB.Saved = True
Next
Application.Quit
End Sub

Hey Franz,

danke für die schnelle Antwort.

Habe deinen Code auch direkt eingefügt, doch leider kommt nachdem ich auf den Button drücke die Fehlermeldung:

Laufzeitfehler 1004:

die informationen können nicht eingefügt werden, da der Berich Kopieren und der Bereich zum Einfügen unterschiedliche Formen und Größen haben.

und diese Zellel wird Markiert:

.UsedRange.PasteSpecial Paste:=xlPasteValues

hoffe du kannst mir da nochmal weiterhelfen:)

Gruß

paddey

Hallo Paddey,

scheinbar gibt es in dem Tabellenblatt verbundene Zellen oder andere problematische Formatierungen.

Möglicherweise stört auch auch ein gesetzter Filter den Kopiervvorgang.

'ersetze die folgende Zeilen
With .Worksheets(1)
.UsedRange.Copy
.UsedRange.PasteSpecial Paste:=xlPasteValues
Application.CutCopyMode = False
End With
'durch'
With .Worksheets(1)
Dim rngZelle As Range
For Each rngZelle In .UsedRange.Cells
If rngZelle.HasFormula Then rngZelle.Value = rngZelle.Value
Next
set rngZelle = Nothing
End With



dann werden alle Zellen in denen sich Formeln befinden einzeln abgearbeitet. Das hat bei mir eigentlich bisher immer funktioniert. Dauert halt etwas länger als die Aktion per Kopieren.

Gruß

Franz

Hey Franz,

vielen vielen Dank :)läuft wunderbar.

Jetzt hätte ich noch eine kleine Frage, um meine Checkliste perfekt zu machen.

Und zwar Kopiere ich mit dieser Formel Text von einem Tabellenblatt in ein anderes:

=WENN('5. Sonstiges'!A5="";"-";'5. Sonstiges'!A5)

Doch leider übernimmt er nicht die Zellenhöhe, auch wenn ich Zeilenumbruch aktiviert habe.

Kennst du da vielleicht noch eine Möglichkeit, wie ich das noch optimieren kann.

Gruß Paddey

Hallo Paddey,

du kannst noch eine Anweisung einfügen, die die Zeilehöhe für die Zeile mit der Formel optimiert.
    Sheets("7. Checkliste").Select
ActiveSheet.AutoFilter.ApplyFilter
Sheets("7. Checkliste").Rows(15).AutoFit 'Nr.der Zeile mit der Formel anpassen!
Sheets("6. Speichern").Select      'Diese Zeile und die nächste sind überflüssig!!
Sheets("7. Checkliste").Select



Dies funktioniert aber nicht mit verbundenen Zellen.

Wenn es nicht stört, dann verbinde die Zellen wieder und verwende die Formatierung

horizontal - zentrieren über Auswahl.

Die Einstellung der korrekten Höhe von verbundenen Zellen entsprechend dem enthaltenen Text ist relativ kompliziert, wenn man es per Makro umsetzen will. Ansonsten setze die Höhe der Zeile manuell auf einen Wert. der meistens passt.

Gruß

Franz

Super danke :) 

Hast mir echt riesig weitergeholfen :)

Vielen vielen DANK

Gruß Paddey

Halli Hallo,

ich habe eine kurze frage. und war geht es darum, dass die spalte a untersucht werden sollen. es sollen alle werte der tabelle (zeilen) kopiert werden, bis in der spalte a der wert 0 ist.

Im Folgenden habe ich eine Beispieldatei drangehangen. https://www.herber.de/bbs/user/...

Hallo VBA-Freaks,

ich suche seit Stunden - finde aber nicht den exakt passenden Code, obwohl es zig Einträge zu dem Thema gibt.

Ich will eine Tabelle per VBA in PDF umwandeln u. drucken. Mein Code:

Application.ActivePrinter = "Adobe PDF auf Ne01:"

ActiveWindow.SelectedSheets.Pri...

Hallo!

Ich habe folgendes Problem:

über einen Commandbutton kopiert der User ein Blatt in eine neue Datei, die so dann gleich geschlossen werden soll. Hier erscheint dann der Dialog "Wollen Sie speichern? - ja - nein - abbrechen -". Ich hätte nun aber gern, dass gleich der Dialog "Speic...

Hallo,

ich meine es gibt die Möglichkeit, die gesamte Arbeitsmappe, also alle Tabellenblätter als ein pdf per Mail  zu versenden .... Doch leider weiß ich nicht mehr wie das geht.

Wer hat ne Lösung?

Gruß und Danke

TOM

Ich habe folgendes Problem, ich möchte ein einzelnes Tabellenblatt aus Excel heraus in eine Pdf Datei mittels command Buttom und Makro/ VBA speichern. Als PDF Prog nutze ich Adobe Acrobat Prof 8.0(bringt nen Converter mit).

THX im Voraus

Hallo Leute, schönen Guten Morgen.

KAnn man folgenden Code so umdrehen, das der Inhalt der Zelle D4 vor dem Blattnamen abgespeichert wird,

ActiveWorkbook.SaveAs SavePath & "\" & ActiveSheet.Name & "  " & ThisWorkbook.Sheets("Bestandsübersicht").Range("D4").Value & "....

